
The tautological ring RH∗(Mg,n) ⊂ H∗(Mg,n) of the moduli space of
stable curves has been studied extensively in the last decades. It has an
explicit additive set of generators and a large (conjecturally complete) set
of relations between these generators. Many operations in this ring (like
intersection products, pullbacks by boundary morphisms ofMg,n, etc.) admit
complete, combinatorial descriptions in terms of the generators.

Given geometrically defined loci in Mg,n, their fundamental class is often
(though not always) contained in the tautological ring. One example are loci
of admissible covers (such as the locus of hyperelliptic or bielliptic curves),
which are tautological in many cases (for low g and n).

In the Sage program admcycles.sage we implement natural operations in
the tautological ring, together with a function identifying formulas for many
cycles of admissible covers in terms of the generating set of RH∗(Mg,n). This
identification works by computing intersection numbers of admissible cover
cycles with tautological classes and by pulling the cycles back via boundary
morphisms. Details will be given in a forthcoming paper. This document
serves as a user manual for admcycles.sage.

We are indebted to Aaron Pixton for letting us use and modify a previous
implementation of operations in the tautological ring by him. The two main
functions of his program that we use are the computation of intersection
numbers of κ and ψ classes and a function computing the generalized Faber-
Zagier relations between the generators of the tautological ring. For details
on the functions that we use see Appendix A.

1. Features of the program

1.1. Data types.

• stable graphs
• tautological classes (on products of moduli spaces of stable curves)
• admissible cover cycles (i.e. loci of curves (C, p1, . . . , pr) admitting
G-Galois covers C → D with ramification points p1, . . . , pr, for fixed
genera, finite group1 G and ramification data)

Date: November 26, 2018.
1Currently, most functions are only implemented for finite cyclic groups G.

1.2. Supported operations.

• intersecting tautological classes (Section 3.1), computing degrees of
tautological zero-cycles (Section 3.2)
• computing a basis of the tautological ring RH2d(Mg,n) and writing

tautological classes in terms of this basis2 (Section 3.3)
• computing pullbacks and pushforwards of tautological classes under

gluing morphism associated to a stable graph (Section 3.4) and
forgetful morphisms (Section 3.2)
• identifying admissible cover cycles in terms of tautological cycles

(Section 4)

1.3. Background computations3.

• handling tautological classes on products of moduli spaces of stable
curves, computing decomposition of diagonals
• intersecting admissible cover cycles with tautological cycles, pulling

them back to the boundary

2. Getting started

Open a command line, go to a directory containing the files admcycles.sage
and DR.py and start sage. Then type

load("admcycles.sage")

The file examples.sage contains several sample computations (with com-
ments), which can be copied and pasted into the command line. We will
describe these computations in more detail in the following.

3. Tautological classes

3.1. Creating tautological classes. There are different ways to enter
tautological classes in the program and depending on the example, some are
more convenient than others.

For boundary divisors as well as ψ, κ and λ-classes, there are predefined
functions.

• sepbdiv(h,A,g,n) gives the pushforward ξ∗[MΓ] of the boundary
gluing map

ξ :MΓ =Mh,A∪{p} ×Mg−h,({1,...,n}\A)∪{p′} →Mg,n,

where A can be a list, set or tuple of numbers from 1 to n. Be careful
that tuples of length 1 must be entered as (a,) in Python, instead
of (a).

2This works in the case where Pixton’s generalized Faber-Zagier relations give all
relations between the generators of the tautological ring.

3These are operations that are implemented and run in the background of some of the
other functions. Currently they are not yet optimized for user access, but once this is done,
we will add a description here.

• irrbdiv(g,n) gives the pushforward ξ∗[Mg−1,n+2] of the boundary
gluing map

ξ :Mg−1,n+2 →Mg,n.

• psiclass(i,g,n) gives the class ψi on Mg,n, defined by

π :Mg,n+1 →Mg,n, σi :Mg,n →Mg,n+1, ψi = c1(σ∗i ωπ).

• kappaclass(a,g,n) gives the (Arbarello-Cornalba) κ-class κa on
Mg,n, defined by

κa = π∗ψ
a+1
n+1.

• lambdaclass(d,g,n) gives the class λd onMg,n, defined as the d-th
Chern class λd = cd(E) of the Hodge bundle E.

When working with a fixed space Mg,n, it is furthermore possible to
specify the desired value of the global variables g and n to avoid giving them
as an argument each time.

These tautological classes can be combined in the usual way by operations
+, -, * and raising to a power ^.

sage: t1=3*sepbdiv(1,(1,2),3,4)-psiclass(4,3,4)^2

sage: g=2;n=1;

sage: t2=-1/3*irrbdiv()*lambdaclass(1)

To enter more complicated classes coming from decorated boundary strata,
it is often convenient to first list all such decorated strata of a certain degree
and then select the desired ones from the list. To give a list of all generators of
Rr(Mg,n) use the function list_tautgens(g,n,r). The list itself is created
by tautgens(g,n,r), from which one can then select the classes.

sage: list_tautgens(2,0,2)

[0] : Graph : [2] [[]] []

Polynomial : 1*(kappa_2^1 )_0

[1] : Graph : [2] [[]] []

Polynomial : 1*(kappa_1^2 )_0

[2] : Graph : [1, 1] [[2], [3]] [(2, 3)]

Polynomial : 1*(kappa_1^1 )_0

[3] : Graph : [1, 1] [[2], [3]] [(2, 3)]

Polynomial : 1*psi_2^1

[4] : Graph : [1] [[2, 3]] [(2, 3)]

Polynomial : 1*(kappa_1^1 )_0

[5] : Graph : [1] [[2, 3]] [(2, 3)]

Polynomial : 1*psi_2^1

[6] : Graph : [0, 1] [[3, 4, 5], [6]] [(3, 4), (5, 6)]

Polynomial : 1*

[7] : Graph : [0] [[3, 4, 5, 6]] [(3, 4), (5, 6)]

Polynomial : 1*

sage: L=tautgens(2,0,2);

sage: t3=2*L[3]+L[4]

sage: t3

Graph : [1, 1] [[2], [3]] [(2, 3)]

Polynomial : 2*psi_2^1

Graph : [1] [[2, 3]] [(2, 3)]

Polynomial : 1*(kappa_1^1 )_0

The output above should be interpreted as follows: each tautclass consists of
a sum of decorated boundary strata (represented by data type decstratum),
which consist of a graph (datatype stgraph) and a polynomial in κ and
ψ-classes (datatype kppoly).

Note: For decorated stratum classes, we have the convention of not
dividing by the order of the automorphism group of the stable graph. Thus
the elements of the list L above are really just pushforwards of κ and ψ
classes on products of moduli spaces Mg,n under appropriate gluing maps.

Let us look at the example of generator L[3] above.

[3] : Graph : [1, 1] [[2], [3]] [(2, 3)]

Polynomial : 1*psi_2^1

Its stable graph is represented by three lists.

(1) The first list [1, 1] are the genera of the vertices, so there are
two vertices, both of genus 1. Note that vertices are numbered by
0, 1, 2, . . ., so in the above case, the vertices are numbers 0 and 1.

(2) the second list gives the legs (that is markings or half-edges) attached
to the vertices, so vertex 0 carries the half-edge 2 and vertex 1 the
half-edge 3,

(3) the third list gives the edges, that is half-edge pairs that are connected;
in the above case, the two half-edges 2 and 3 form an edge, connecting
the two vertices

If we wanted to enter this stgraph manually, we could use its constructor as
follows:

sage: stgraph([1,1],[[2],[3]],[(2,3)])

[1, 1] [[2], [3]] [(2, 3)]

The polynomial in κ and ψ is 1*psi_2^1 in this case, so the half-edge 2 on
the second vertex carries a ψ-class. For the generator L[4] the polynomial
looks like 1*(kappa_1^1 )_0, meaning that vertex 0 carries a class κ1

1 = κ1.
Finally, it is possible to manually enter tautological classes by using the

constructors of the classes tautclass, decstratum and so on. We refer to
comments in the source code admcycles.sage for details on this.

3.2. Basic operations. Apart from the usual arithmetic operations, we
can take forgetful pushforwards and pullbacks of tautological classes and
also compute the degree of tautological zero-cycles. In particular, we can
compute intersection numbers. Below, for the forgetful map π :M1,3 →M1,2

forgetting the marking 3 we verify the relations

π∗ψ
2
3 = κ1, π

∗ψ2 = ψ2 −D0,{2,3},

and we compute the intersection number

〈τ0τ1τ2〉1,3 =

∫
M1,3

ψ0
1ψ2ψ

2
3 = 1/12.

We check that it agrees with the prediction 〈τ0τ1τ2〉1,3 = 〈τ0τ2〉1,2 + 〈τ2
1 〉1,2

by the string equation.

sage: s1=psiclass(3,1,3)^2

sage: s1.forgetful_pushforward([3])

Graph : [1] [[1, 2]] []

Polynomial : 1*(kappa_1^1 )_0

sage: s2=psiclass(2,1,2)

sage: s2.forgetful_pullback([3])

Graph : [1] [[1, 2, 3]] []

Polynomial : 1*psi_2^1

Graph : [1, 0] [[1, 4], [5, 3, 2]] [(4, 5)]

Polynomial : -1*

sage: s3=psiclass(2,1,3)*psiclass(3,1,3)^2

sage: s3.evaluate()

1/12

sage: s4=psiclass(2,1,2)^2+psiclass(1,1,2)*psiclass(2,1,2)

sage: s4.evaluate()

1/12

3.3. A basis of the tautological ring and tautological relations. The
function generating_indices(g,n,r) computes the indices (with respect to
Pixton’s generating set) of a basis of Rr(Mg,n), assuming that the generalized
Faber-Zagier relations between the additive generators give a complete set of
relations between them. The function Tautvecttobasis converts a vector
with respect to the whole generating set into a vector in this basis. The
function tautclass.toTautbasis(g,n,r) converts a tautclass into such
a vector.

Continuing the example from Section 3.1 we see:

sage: generating_indices(2,0,2)

[0, 1]

sage: t3.toTautbasis(2,0,2)

(-48, 22)

This means that generators L[0], L[1] form a basis of R2(M2) and the
tautclass t3=2*L[3]+L[4] is equivalent to -48*L[0]+22*L[1].

We can also directly verify tautological relations using the built-in function
is_zero of tautclass. Below we verify the known relation κ = ψ − δ0 ∈
R1(M1,n) for n = 4. Here ψ is the sum of all ψi and δ0 is the sum of all
separating boundary divisors, i.e. those having a genus 0 component. For
this, we list all stable graphs with one edge via list_strata(g,n,1). We
exclude the graph gamma with a self-loop by requiring that the number of
vertices gamma.numvert() is at least 2. Then we can convert these graphs
to tautclases by using to_tautclass.

sage: g=1;n=4;

sage: bgraphs=[bd for bd in list_strata(g,n,1) if bd.numvert()>1]

sage: del0=sum([bd.to_tautclass() for bd in bgraphs])

sage: psisum=sum([psiclass(i) for i in range(1,n+1)])

sage: rel=kappaclass(1)-psisum+del0

sage: rel.is_zero()

True

In practice, much of the time in some computations is spent on calculat-
ing generalized Faber-Zagier relations between tautological cycles on Mg,n.
However, once computed, the relations can be saved to a file and be reloaded
in a later session using the functions save_FZrels() and load_FZrels().
Careful: the function save_FZrels() creates (and overwrites previous ver-
sion of) a file geninddb.pkl which depending on the previous computations
can be quite large.

3.4. Pulling back tautological classes to the boundary. Below we pull
back a generator of R2(M4) to the boundary divisor with genus partition
4 = 2 + 2. This produces an element of type prodtautclass, a tautological
class on a product of moduli spaces, in this caseM2,1×M2,1. Two elements
on the same product of spaces can be added and multiplied and further
operations like pushforwards under (partial) gluing maps are supported. We
will describe this in more detail in a later version of this manual (but some
explanation is already provided in comments in the code of admcycles.sage).

In our case, we want to express the pullback to M2,1 ×M2,1 in terms of

a basis of H2(M2,1 ×M2,1) obtained from the preferred bases of the factors

H∗(M2,1) given by generating_indices. We can either represent the result
as a list of matrices (giving the coefficients in the tensor product bases) or
as a combined vector (using the option vecout=true).

sage: bdry=stgraph([2,2],[[1],[2]],[(1,2)])

sage: generator=tautgens(4,0,2)[3]; generator

Graph : [1, 3] [[2], [3]] [(2, 3)]

Polynomial : 1*psi_3^1

sage: pullback=bdry.boundary_pullback(generator);

sage: pullback.totensorTautbasis(2)

[

[-3]

[ 1]

[0 0 0] [-3]

[0 0 0] [ 7]

[-3 1 -3 7 1], [0 0 0], [ 1]

]

sage: pullback.totensorTautbasis(2,vecout=true)

(-3, 1, -3, 7, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, -3, 1, -3, 7, 1)

4. Admissible cover cycles

4.1. Shortcut: Hyperelliptic and bielliptic cycles. Before we go into
details of how to specify general admissible cover cycles, let us mention the
important cases of hyperelliptic and bielliptic cycles.

Given g ≥ 0, 0 ≤ n ≤ 2g + 2 and m ≥ 0 with 2g − 2 + n + 2m > 0, we
have the locus Hg,n,2m ⊂Mg,n+2m of curves (C, p1, . . . , pn, q1, q

′
1, . . . , qm, q

′
m)

such that C is hyperelliptic with p1, . . . , pn fixed points of the hyperellip-
tic involution and the pairs qi, q

′
i being exchanged by this involution. An

analogous definition gives the locus Bg,n,2m ⊂Mg,n+2m of bielliptic curves
with n fixed points and m pairs of points forming orbits under the bielliptic
involution.

Then the fundamental class of the (reduced) loci Hg,n,2m and Bg,n,2m

can (in many cases) be computed by the functions Hyperell(g,n,m) and
Biell(g,n,m) of our program.

As an example, we compute the class [H3] ∈ A1(M3) and verify that we
obtain the known result

[H3] = 9λ− δ0 − 3δ1,

where δ0 is the class of the divisor of irreducible nodal curves and δ1 is the
divisor of curves with a separating node between a genus 1 and a genus 2
component.

sage: H=Hyperell(3,0,0)

sage: H.toTautbasis()

(3/4, -9/4, -1/8)

sage: g=3; n=0;

sage: H2=9*lambdaclass(1)-(1/2)*irrbdiv()-3*sepbdiv(1,())

sage: H2.toTautbasis()

(3/4, -9/4, -1/8)

Here we need to divide irrbdiv() by two, the degree of the corresponding
gluing map.

4.2. Creating and identifying admissible cover cycles. In general, an
admissible cover cycle is specified by a genus, a group as well as monodromy
data. Currently, intersections are only implemented for cyclic groups. Below
we will study bielliptic curves in genus 2, which are double covers of elliptic
curves branched over two points. As a first step we enter the ramification
data.

sage: G=PermutationGroup([(1,2)])

sage: list(G)

[(), (1,2)]

sage: H=HurData(G,[G[1],G[1]])

The function HurData takes the group G as the first argument and as the
second a list of group elements α ∈ G, each of which corresponds to the
G-orbit of some marking p ∈ C. Here α is a generator of the stabilizer of
p under the group action G y C, which gives the monodromy around p.
In other words, the natural action of the stabilizer Gp = 〈α〉 on a tangent
vector v ∈ TpC is given by

α.v = exp(2πi/ord(h))v.

Thus in the example above, we have two markings, both with stabilizer
generated by G[1]=(1,2) which acts by multiplication of −1 on the tangent
space.

To identify the admissible cover cycle (inside the moduli space Mg,n with
n the total number of marked points from the ramification data) in terms of
tautological classes, one can use the function Hidentify.

It pulls back the admissible cover cycle to all boundary divisors and
(recursively) identifies the pullback itself in terms of tautological classes. It
compares this pullback to the pullback of a basis of the tautological ring.
Often this pullback map is injective in cohomology such that one can then
write the admissible cover cycle in terms of the basis using linear algebra.
Sometimes, it is necessary to additionally intersect with some monomials in
κ and ψ-classes.

To apply Hidentify one gives the genus and the monodromy data as ar-
guments. The standard output format is an instance of the class tautclass.
For those users familiar with Aaron Pixton’s implementation of the tauto-
logical ring, there is the option vecout=true which returns instead a vector
with respect to the generating set of the tautological ring provided by this
program.

sage: vbeta=Hidentify(2,H,vecout=true)

sage: vector(vbeta)

(517/4, -33, 11/4, 243/4, -125/4, 15/2, 41/4, 125, 99/4, -41,

-1137/4, -285/4, 0, 0, 0, 0, 0, 0, -57/8, -3/8, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0,

0, 0, 0, 0, 0, 0)

The output above means specifically, that inside M2,2 the locus of bielliptic
curves with the two points fixed by the involution being the marked points
is given (in the generating set gens=all_strata(2,3,(1,2)) produced by
Pixton’s program) as 517/4 · gens[0]− 33 · gens[1] + . . ..

If we instead wanted to have bielliptic curves with two marked fixed-
points of the involution and one pair of markings that are exchanged by the
involution, we would need to use the monodromy data

sage: H2=HurData(G,[G[1],G[1],G[0]])

in which case Hidentify(2,H2) would live inside R4(M2,4).
If we only want to remember a subset of the markings, we can use

the optional parameter marking to give this subset. For instance, the
command Hidentify(2,H,markings=[]) would give the pushforward of
Hidentify(2,H) in M2,2 to the space M2 under the forgetful morphism
(see also Section 4.3).

4.3. Example: Specifying and Identifying [B2] by hand. The locus
B2 ⊂ M2 of bielliptic curves is a divisor. A bielliptic genus 2 curve is
ramified over two points. In the following we want to use the methods of the
previous section to identify its cycle class.

Now when treating admissible cover cycles in general, our program a priori
handles the cycle where all possible ramification points are marked. In this
case, this is the cycle [B2,2,0] ∈ R3(M2,2) of bielliptic curves C with the two
ramification points p1, p2 marked. By specifying markings=[] when calling
Hidentify, we tell it to remember none of the markings, in other words to
push forward under the map π :M2,2 →M2 forgetting the markings.

sage: G=PermutationGroup([(1,2)])

sage: H=HurData(G,[G[1],G[1]])

sage: Biell=Hidentify(2,H,markings=[])

sage: Biell.toTautbasis(2,0,1)

(30, -9)

We want to compare the result with the known formulas for [B2]. For δ0 the
class of the irreducible boundary of M2 and δ1 the class of the boundary

divisor with genus-splitting (1, 1), it is known that [B2] = 3
2δ0 + 6δ1. If we

want to enter this combination of δ0 and δ1, we have to be careful about
conventions, though: the corresponding gluing maps ξ : M1,2 → M2 and

ξ′ :M1,1 ×M1,1 →M2 both have degree 2. This corresponds to the fact
that the associated stable graphs both have an automorphism group of order
2. Hence we have to divide by a factor of two and obtain:

sage: g=2;n=0

sage: Biell2=3/4*irrbdiv()+ 3*sepbdiv(1,())

sage: Biell2.toTautbasis(2,0,1)

(15/2, -9/4)

We see that up to a factor of 4 the two vectors (30, -9) and (15/2, -9/4)

agree. Where does this factor come from?
For this recall that the cycle Biell above is equal to π∗[B2,2,0]. Since for

the generic bielliptic curve C there are two choices of orderings for marking
p1, p2, this explains a factor of 2. On the other hand, the hyperelliptic
involution σ : C → C on C exchanges p1 and p2. Thus σ ∈ Aut(C), but
σ /∈ Aut(C, p1, p2). This missing automorphism factor explains another
factor of 2 in the pushforward under π, so in fact [B2] = 1

4π∗[B2,2,0].
Note that since the cycles of bielliptic loci are implemented via the function

Biell, we could have taken a shortcut above.

sage: B=Biell(2,0,0); B.toTautbasis()

(15/2, -9/4)

As an application, we can check the Hurwitz-Hodge integral∫
[B2,2,0]

λ2λ0 =

∫
π∗[B2,2,0]

λ2 =
1

48

predicted by [JPT11].

sage: (Biell*lambdaclass(2,2,0)).evaluate()

1/48

In principle the corresponding integrals for g = 3, 4 can also be verified like
this, but the amount of time and memory needed grows drastically.

Appendix A. Functions from Aaron Pixton’s program

Many functions of our program rely in essential ways on a previous imple-
mentation by Aaron Pixton. It is imported in the form of the precompiled
file DRpython.pyc. The original code (slightly modified by us) can be down-
loaded from our homepage.

Below we give an overview which functions we use.

• all_strata/ num_strata - This computes a list/the number of all
decorated stratum classes in a given degree on Mg,n. For the sake of
compatibility, we chose to use the same order in our own program.
• FZ_matrix - Computes the generalized Faber-Zagier relations be-

tween the decorated stratum classes in a given degree. We use
this to identify tautological classes in terms of a basis and to verify
tautological relations.
• socle_formula - Evaluates the integral of a monomial in κ and
ψ-classes. We use this when computing intersection numbers, both
for tautological classes and for the intersection of a tautological class
with an admissible cover cycle.
• pairing_submatrix - Computes a submatrix of the pairing matrix

between complementary degrees in the tautological ring. We imple-
mented our own version of the intersection in the tautological ring,
so using socle_formula we could compute these numbers ourselves.
However, for small g, n the implementation by Pixton is faster, so
currently we use his version of the function.
• We use a few more functions for converting tautological classes and

relations from Pixton’s program to our own.

land

E-mail address: johannes.schmitt@math.ethz.ch

Deparment of Mathematics, Humboldt-Universität zu Berlin, Rudower
Chaussee 25, Room 1.415, 12489 Berlin, Germany

E-mail address: jasonvanzelm@outlook.com


	1. Features of the program
	1.1. Data types
	1.2. Supported operations
	1.3. Background computationsThese are operations that are implemented and run in the background of some of the other functions. Currently they are not yet optimized for user access, but once this is done, we will add a description here.

	2. Getting started
	3. Tautological classes
	3.1. Creating tautological classes
	3.2. Basic operations
	3.3. A basis of the tautological ring and tautological relations
	3.4. Pulling back tautological classes to the boundary

	4. Admissible cover cycles
	4.1. Shortcut: Hyperelliptic and bielliptic cycles
	4.2. Creating and identifying admissible cover cycles
	4.3. Example: Specifying and Identifying [B2] by hand

	Appendix A. Functions from Aaron Pixton's program
	References

